Soluzioni del compito di

Metodi Matematici e Calcolo per Ingegneria

del 3 novembre 2005

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> t*sin(t) ;
```

$$f := t \to t\sin(t)$$

Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(t/3) ;
  f2 := t -> f(t/2)*exp(-2*t) ;
  f3 := t -> D(D(D(f)))(t) ;
```

$$f1 := t \to f\left(\frac{1}{3}\,t\right)$$

$$f2 := t \to f\left(\frac{1}{2}\,t\right)\,\mathbf{e}^{(-2\,t)}$$

$$f3 := t \to \mathrm{D}(\mathrm{D}(\mathrm{D}(f)))(t)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{2\,s}{(s^2+1)^2}$$

$$\frac{18\,s}{(9\,s^2+1)^2}$$

$$\frac{8\,(s+2)}{(4\,s^2+16\,s+17)^2}$$

$$-\frac{2\,(2\,s^2+1)}{(s^2+1)^2}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente equazione differenziale

```
> ode := diff(y(x),x,x,x)-diff(y(x),x)=x ;
```

$$ode := \left(\frac{d^3}{dx^3}\,y(x)\right) - \left(\frac{d}{dx}\,y(x)\right) = x$$

Con dato iniziale

```
> y0, yp0, ypp0 := 1,-1,0 ;
```

$$y0,\, yp0,\, ypp0 := 1, -1, 0$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s^3\ laplace(y(x), x, s) - \mathrm{D}^{(2)}(y)(0) - s\ \mathrm{D}(y)(0) - s^2\ y(0) - s\ laplace(y(x), x, s) + y(0) = \frac{1}{s^2}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x), x, s) = \frac{\frac{1}{s^2} + \mathrm{D}^{(2)}(y)(0) + s\ \mathrm{D}(y)(0) + s^2\ y(0) - y(0)}{s^3 - s}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := subs( y(0)=y0,D(y)(0)=yp0,D(D(y))(0)=ypp0,rhs(lode)) ;
```

$$ly := \frac{\frac{1}{s^2} - 1 - s + s^2}{s^3 - s}$$

Espansione in fratti semplici

```
> convert(ly, parfrac, s);
```

$$-\frac{1}{s^3}+\frac{1}{s+1}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := \mathrm{e}^{(-t)} - \frac{1}{2}t^2$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> yp, zp, wp := diff(y(x),x),diff(z(x),x),diff(w(x),x);
```

$$yp, zp, wp := \frac{d}{dx}y(x), \frac{d}{dx}z(x), \frac{d}{dx}w(x)$$

```
> ode1 :=   yp + zp + wp = 1   ;
  ode2 :=        zp + wp = x   ;
  ode3 := - yp      + wp = x^2 ;
```

$$ode1 := \left(\frac{d}{dx}y(x)\right)+\left(\frac{d}{dx}z(x)\right)+\left(\frac{d}{dx}w(x)\right)=1$$

$$ode2 := \left(\frac{d}{dx}z(x)\right)+\left(\frac{d}{dx}w(x)\right)=x$$

$$ode3 := -\left(\frac{d}{dx}y(x)\right)+\left(\frac{d}{dx}w(x)\right)=x^2$$

Con dato iniziale

```
> y0, z0, w0 := 3, 2, 1 ;
```

$$y0, z0, w0 := 3, 2, 1$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo le equazioni differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
  sode3 := laplace(ode3,x,s) ;
```

$$sode1 := s\ laplace(y(x), x, s) - y(0) + s\ laplace(z(x), x, s) - z(0) + s\ laplace(w(x), x, s) - w(0) = \frac{1}{s}$$

$$sode2 := s\ laplace(z(x), x, s) - z(0) + s\ laplace(w(x), x, s) - w(0) = \frac{1}{s^2}$$

$$sode3 := -s\ laplace(y(x), x, s) + y(0) + s\ laplace(w(x), x, s) - w(0) = \frac{2}{s^3}$$

Risolvo la equazione per y(s), z(s)

```
> ys,zs,ws := laplace(y(x),x,s),laplace(z(x),x,s),laplace(w(x),x,s);
```

$$ys, zs, ws := laplace(y(x), x, s), laplace(z(x), x, s), laplace(w(x), x, s)$$

```
> RES := solve({sode1,sode2,sode3},{ys,zs,ws});
```

$$RES := \left\{ laplace(y(x), x, s) = \frac{-1 + y(0)\, s^2 + s}{s^3}, laplace(z(x), x, s) = \frac{-2 + 2\, s - s^2 + z(0)\, s^3}{s^4}, laplace(w(x), x, s) = \frac{2 - s + s^2 + w(0)\, s^3}{s^4} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL := subs(RES,y(0)=y0,z(0)=z0,w(0)=w0,<ys,zs,ws>);
```

$$SOL := \begin{bmatrix} \frac{-1 + 3\, s^2 + s}{s^3} \\ \frac{-2 + 2\, s - s^2 + 2\, s^3}{s^4} \\ \frac{2 - s + s^2 + s^3}{s^4} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
  ww := invlaplace(SOL[3],s,x) ;
```

$$yy := -\frac{1}{2} x^2 + 3 + x$$

$$zz := -\frac{1}{3} x^3 + x^2 - x + 2$$

$$ww := \frac{1}{3} x^3 - \frac{1}{2} x^2 + x + 1$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], parfrac, s);
  convert(SOL[2], parfrac, s);
  convert(SOL[3], parfrac, s);
```

$$\frac{3}{s}+\frac{1}{s^2}-\frac{1}{s^3}$$

$$-\frac{2}{s^4}+\frac{2}{s}-\frac{1}{s^2}+\frac{2}{s^3}$$

$$\frac{2}{s^4}+\frac{1}{s}+\frac{1}{s^2}-\frac{1}{s^3}$$

Soluzione di ricorrenza con trasformata zeta

```
> expand((z-1)^3);
```

$$z^3-3z^2+3z-1$$

```
> restart:
```

Risolvere la seguente ricorrenza

```
> RIC := f(n+3) = 3*f(n+2)- 3*f(n+1) + f(n) - 1 ;
```

$$RIC := f(n+3)=3\,f(n+2)-3\,f(n+1)+f(n)-1$$

Con dato iniziale

```
> INI := f(0)=0,f(1)=1,f(2)=1 ;
```

$$INI := f(0)=0, f(1)=1, f(2)=1$$

Usando le primitive si maple:

```
> rsolve({RIC,INI}, f(k));
```

$$2\,(k+1)\left(\frac{1}{2}k+1\right)-1-(k+1)\left(\frac{1}{2}k+1\right)\left(\frac{1}{3}k+1\right)$$

```
> simplify(%);
```

$$\frac{7}{6}k-\frac{1}{6}k^3$$

Usando la Z-trasformata (occhio negli appunti abbiamo Z(f) = sum(f(n)*z^n mentre maple usa Z(f) = sum(f(n)*w^{-n}

Quindi per confrontare le trasformate bisogna trasfromare w -> 1/z;

```
> zRIC := ztrans(RIC,n,w):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRIC)) ;
```

$$-\frac{-f(z)+f(0)+f(1)\,z+f(2)\,z^2}{z^3}=$$
$$\frac{-3\,f(z)+6\,f(z)\,z+3\,f(0)-6\,f(0)\,z+3\,f(1)\,z-3\,f(1)\,z^2-4\,f(z)\,z^2+3\,f(0)\,z^2+f(z)\,z^3+z^2}{z^2\,(-1+z)}$$

Ricavo f(w) [f(1/z)]

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,w)):
```

```
simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhs)) ;
```

$$f(z) = -\frac{-f(0) + 4f(0)z - f(1)z + 4f(1)z^2 - f(2)z^2 + f(2)z^3 - 6f(0)z^2 - 3f(1)z^3 + 3f(0)z^3 + z^3}{(-1+z)(-1+3z-3z^2+z^3)}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhsINI)) ;
```

$$f(z) = \frac{(1-3z+z^2)z}{(-1+z)(-1+3z-3z^2+z^3)}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$f(z) = -\frac{2}{(-1+z)^3} + \frac{1}{-1+z} - \frac{1}{(-1+z)^4}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,w,k) ;
```

$$f(k) = \frac{7}{6}k - \frac{1}{6}k^3$$

Soluzione di un sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`,
`+`, `.`, D, Vector, diff, int, limit, series
```

Sistema non lineare

```
> f := 2*x + y   + x*y + 1 ;
  g :=   x - 2*y - x*y - 2 ;
```

$$f := 2x + y + xy + 1$$

$$g := x - 2y - xy - 2$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\left\{x = \frac{-4}{3}, y = -5\right\}, \{y = -1, x = 0\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} 2+y & 1+x \\ 1-y & -2-x \end{bmatrix}$$

Schema di Newton

```
> Newton_update := simplify(<x,y>-J^(-1).<f,g>);
```

$$Newton\_update := \frac{x\,(y+1)}{5+3\,x+y}\,e_x + \frac{-y+3\,x\,y-5}{5+3\,x+y}\,e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1} = \frac{x_k\,(1+y_k)}{5+3\,x_k+y_k}$$

$$y_{k+1} = \frac{-y_k+3\,x_k\,y_k-5}{5+3\,x_k+y_k}$$

Tre iterate a partire da (1,2)

```
> x[0],y[0]:= 1,2 ;
```

$$x_0, y_0 := 1, 2$$

Prima iterata

```
> x[1] := evalf(subs(x=x[0],y=y[0],Newton_update[1])) ;
  y[1] := evalf(subs(x=x[0],y=y[0],Newton_update[2])) ;
```

$$x_1 := 0.3000000000$$

$$y_1 := -0.1000000000$$

Seconda iterata

```
> x[2] := evalf(subs(x=x[1],y=y[1],Newton_update[1])) ;
  y[2] := evalf(subs(x=x[1],y=y[1],Newton_update[2])) ;
```

$$x_2 := 0.04655172414$$

$$y_2 := -0.8603448276$$

Terza iterata

```
> x[3] := evalf(subs(x=x[2],y=y[2],Newton_update[1])) ;
  y[3] := evalf(subs(x=x[2],y=y[2],Newton_update[2])) ;
```

$$x_3 := 0.001519214205$$

$$y_3 := -0.9954423577$$

Problema di Minimo Vincolato

```
> restart:
  with(LinearAlgebra):
```

```
with(Optimization):
with(VectorCalculus):
```

```
Warning, the names `&x`, CrossProduct and DotProduct have been rebound

Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`, `+`, `.`, D, Vector, diff, int, limit, series
```

Minimizzare la seguente funzione

```
> f := x-y+z^3;
```

$$f := x - y + z^3$$

Soggetta ai seguenti vincoli

```
> v := [x+y-z=3,x+z=0] ;
```

$$v := [x + y - z = 3, x + z = 0]$$

Soluzione con le primitive Maple

```
> Minimize(f, v );
```

$$[-5., [x = -1.00000000000003820, y = 5.00000000000007726, z = 1.00000000000003864]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```

$$v1 := x + y - z - 3$$

$$v2 := x + z$$

```
> g := f - lambda*v1 - mu*v2 ;
```

$$g := x - y + z^3 - \lambda\,(x + y - z - 3) - \mu\,(x + z)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```

$$F := (1 - \lambda - \mu)\,\bar{e}_x + (-1 - \lambda)\,\bar{e}_y + (3\,z^2 + \lambda - \mu)\,\bar{e}_z + (-x - y + z + 3)\,\bar{e}_\lambda + (-x - z)\,\bar{e}_\mu$$

Soluzioni del sistema non lineare

```
> RES := solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,mu}) ;
```

$$RES := \{x = -1, z = 1, \mu = 2, \lambda = -1, y = 5\}, \{y = 1, z = -1, \mu = 2, \lambda = -1, x = 1\}$$

Prima soluzione

```
> RES[1];
```

$$\{x = -1, z = 1, \mu = 2, \lambda = -1, y = 5\}$$

Seconda soluzione

```
> RES[2];
```

$$\{y = 1, z = -1, \mu = 2, \lambda = -1, x = 1\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
  Hv1 := Hessian(v1,[x,y,z]):
  Hv2 := Hessian(v2,[x,y,z]):
  Hf, Hv1, Hv2 ;
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 6z \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```

$$JH := \begin{bmatrix} 1 & 1 & -1 \\ 1 & 0 & 1 \end{bmatrix}$$

$$NH := \left\{ \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix} \right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES[1],lambda);
  mu1     := subs(RES[1],mu);
```

$$\lambda 1 := -1$$

$$\mu 1 := 2$$

```
> Hf1 := simplify(subs(RES[1],Hf - lambda1. Hv1 - mu1. Hv2)) ;
```

$$Hf1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 6 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES[1],op(NH)) ;
```

$$Z1 := \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix}$$

E' positivo per ogni alpha, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$6\,\alpha^2$$

```
> subs(RES[1],f);
```

$$-5$$

Controllo minimo/massimo locale secondo punto

```
> lambda2 := subs(RES[2],lambda);
  mu2     := subs(RES[2],mu);
```

$$\lambda 2 := -1$$

$$\mu 2 := 2$$

```
> Hf1 := simplify(subs(RES[2],Hf - lambda1. Hv1 - mu1. Hv2)) ;
```

$$Hf1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & -6 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES[2],op(NH)) ;
```

$$Z1 := \begin{bmatrix} -1 \\ 2 \\ 1 \end{bmatrix}$$

E' negativo per ogni alpha, quindi è un massimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$-6\,\alpha^2$$

```
> subs(RES[2],f);
```

$$-1$$

```
>
>
```